

**Tasks:**

**Write queries to retrieve data on entities of the University class (<http://dbpedia.org/ontology/University>) in dbpedia according to the following specifications (All queries are limit to 10 results).**

1. Get the names of all universities that has more than 4000 faculties and has campus in urban area in alphabetical order.
2. Get the names of all alumni from a university chosen by yourself with names in alphabetical order.
3. Get the names of all universities with more than 100 alumni sorted by the numbers of alumni (ascending).
4. Get the names of all universities that has more graduate students than undergraduate students sorted by the differences between number of graduate students and undergraduate students.
5. Get the names of all universities that locates in three largest states of the United States in reverse alphabetical order.
6. Get the names of all universities that has “of” in its name in ascending order in Oregon.

**Query1:**

```

PREFIX dbp: <http://dbpedia.org/property/>
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

select distinct ?u_names
where {
  ?universities a dbo:University;
    rdfs:label ?u_names;
    dbp:campus|dbo:campus ?campus;
    dbo:facultySize ?fs.
  FILTER (ucase(str(?campus))="URBAN" || ?campus=<http://dbpedia.org/resource/Urban_area>).
  FILTER (?fs > 4000) .
}

ORDER BY ASC (?u_names)

LIMIT 10

```

**Results:**

u_names
"Akaki Tsereteli State University"@en
"Baylor College of Medicine"@en
"Cairo University"@en
"Central South University"@en
"Central University of Venezuela"@en
"City Colleges of Chicago"@en
"Columbia University College of Physicians and Surgeons"@en
"Florida State University"@en
"Guizhou University"@en
"Harvard University"@en

Ps. ?campus=[http://dbpedia.org/resource/Urban\\_area](http://dbpedia.org/resource/Urban_area) <=> ?campus=dbp:Urban\_area

**Explanation:**

Select 10 distinct university names which are universities, have name property, campus property or campus ontology, and facultySize ontology properties in alphabetical order. Upper case of string of campus equal to “URBAN” or campus equal to “<http://dbpedia.org/resource/Urban\_area>”. Their faculty size > 4000.

**Query2:**

```
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
```

```
PREFIX dbp: <http://dbpedia.org/property/>
```

```
PREFIX dbo: <http://dbpedia.org/ontology/>
```

```
select distinct ?person_names
```

```
where {
```

```
  ?person a dbo:Person;
```

```
    foaf:name ?person_names;
```

```
    dbp:almaMater ?school.
```

```
  ?school a dbo:University;
```

```
    foaf:name ?university_name .
```

```
  FILTER (str(?university_name)="University of Southern California") .
```

```
}
```

```
ORDER BY ASC (?person_names)
```

```
LIMIT 10
```

**Results:**

person_names
"Aaron McCreary"@en
"Albert Magnoli"@en
"Allen Siple"@en
"Andrew Getty"@en
"Anthony Marra"@en
"Armen Margarian"@en
"B. Wayne Hughes, Jr."@en
"Bernard Zimmerman"@en
"Braden Danner"@en
"Brice Taylor"@en

**Explanation:**

Select 10 distinct names of person which is a person object, have name property, attend a university school which has a name of “University of Southern California” in alphabetical order.

**Query3:**

```
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX dbp: <http://dbpedia.org/property/>
PREFIX dbo: <http://dbpedia.org/ontology/>

select distinct ?u_name
where {
  ?school a dbo:University;
    rdfs:label ?u_name;
    dbp:alumni ?alumni.
  FILTER (lang(?u_name) = "en")
  FILTER (?alumni > 100.0 && (datatype(?alumni)=xsd:integer || datatype(?alumni)=xsd:double) ).
}

ORDER BY DESC (?alumni)
Limit (10)
```

**Results:**

u_name
"University of North Texas"@en
"St. Albert's College"@en
"Srinakharinwirot University"@en
"University of Cincinnati"@en
"National University of Ireland"@en
"Kwansei Gakuin University"@en
"Tunku Abdul Rahman University College"@en
"University of Florida College of Liberal Arts and Sciences"@en
"Ryerson University"@en
"Cooperative University of Colombia"@en

**Explanation:**

Select 10 distinct English name of schools which are universities with properties of “alumni” and more than > 100 alumnus. The data type of alumni is either integer or double. University names sort by descending order of the number of alumni.

**Query4:**

```
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX dbp: <http://dbpedia.org/property/>
PREFIX dbo: <http://dbpedia.org/ontology/>

select distinct ?u_name
where {
  ?school a dbo:University;
    rdfs:label ?u_name;
    dbo:numberOfGraduateStudents ?n_graduate;
    dbo:numberOfUndergraduateStudents ?n_undergraduate.
  FILTER (lang(?u_name) = "en").
}

ORDER BY DESC (?n_graduate - ?n_undergraduate)

Limit (10)
```

**Results:**

u_name
"Wolfson College, Cambridge"@en
"St Cross College, Oxford"@en
"Hughes Hall, Cambridge"@en
"St Edmund's College, Cambridge"@en
"Homerton College, Cambridge"@en
"Lucy Cavendish College, Cambridge"@en
"Harris Manchester College, Oxford"@en
"St Stephen's House, Oxford"@en
"Lincoln College, Oxford"@en
"Merton College, Oxford"@en

**Explanation:**

Select 10 distinct English name of schools which are universities with properties of “number of graduate students” and “number of undergraduate students”. Sorting by descending order of the differences that number of graduate students minus undergraduate students.

**Query5:**

```
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX dbp: <http://dbpedia.org/property/>
PREFIX dbo: <http://dbpedia.org/ontology/>

select distinct ?u_name
where {
  ?school a dbo:University;
    rdfs:label ?u_name;
    dbo:state ?big_state.
  FILTER (lang(?u_name) = 'en')
  {
    select distinct ?big_state
    where {
      ?big_state a dbo:PopulatedPlace.
      ?big_state dbo:areaTotal ?area.
      ?big_state dbo:demonym ?state.
      ?big_state dbo:country <http://dbpedia.org/resource/United_States>.
    }
    ORDER BY DESC (?area)
    Limit (3)
  }
}

ORDER BY DESC (?u_name)
Limit (10)
```

**Results:**

u_name
"Ziegler School of Rabbinic Studies"@en
"Zaytuna College"@en
"Yeshiva Ohr Elchonon Chabad/West Coast Talmudical Seminary"@en
"Woodland Community College"@en
"William Jessup University"@en
"William Carey International University"@en
"Wiley College"@en
"Wiess School of Natural Sciences"@en
"Whittier College"@en
"Whitacre College of Engineering"@en

**Explanation:**

Select 10 distinct English name of schools which are universities and have a state property which is top 3 total area of populated place with demonym property in the country of the United States.

**Query6:**

```
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
```

```
PREFIX dbp: <http://dbpedia.org/property/>
```

```
PREFIX dbo: <http://dbpedia.org/ontology/>
```

```
select distinct ?university_name
```

```
where {
```

```
?school a dbo:University;
```

```
    rdfs:label ?university_name;
```

```
    dbo:state ?state.
```

```
FILTER (CONTAINS(str(?state), "Oregon")).
```

```
FILTER (CONTAINS(str(?university_name), " of ")) .
```

```
}
```

```
ORDER BY ASC (?university_name)
```

```
Limit (10)
```

**Results:**

university_name
"Art Institute of Portland"@en
"Charles H. Lundquist College of Business"@en
"College of Osteopathic Medicine of the Pacific, Northwest"@en
"Le Cordon Bleu College of Culinary Arts in Portland"@en
"Maseeh College of Engineering and Computer Science"@en
"National University of Natural Medicine"@en
"OGI School of Science and Engineering"@en
"Oregon College of Art and Craft"@en
"Oregon College of Oriental Medicine"@en
"Oregon Health & Science University School of Dentistry"@en

**Explanation:**

Select 10 distinct names of university which are universities, own name property contains with “ of “ in the school name, and have a state ontology which contains string of “Oregon” in ascending order of university names.



